

% depth-first search

```
depthfirst(Start,[Start]) :- goal(Start),!.
depthfirst(Start,[Start|Rest]) :-
    move(Start, Next),
    depthfirst(Next, Rest).
```

% depth-first acyclic search

```
depthfirst_acyclic(Start,Solution) :-
    depthfirst_aux(Start,[Start],Temp_Solution),
    reverse(Temp_Solution, Solution).

depthfirst_aux(Start, Sol, Sol) :- goal(Start), !.
depthfirst_aux(Node, Path, Solution) :-
    move(Node,Next),
    \+ member(Next, Path),
    depthfirst_aux(Next, [Next|Path], Solution).
```

% depth-first depth-limited search

```
depthfirst_DL(Start,[Start],_) :- goal(Start),!.
depthfirst_DL(Start,[Start|Rest], Max) :-
    Max > 0,
    move(Start, Next),
    Next_Max is Max -1,
    depthfirst_DL(Next, Rest, Next_Max).
```

% depth-first depth-limited acyclic search

```
depthfirst_DL_acyclic(Start,Solution, Max) :-
    depthfirst_DL_aux(Start,[Start],Temp_Solution,Max),
    reverse(Temp_Solution, Solution).
depthfirst_DL_aux(Start,Solution, Solution, _) :- goal(Start),!.
depthfirst_DL_aux(Start,Path,Solution, Max) :-
    Max > 0,
    move(Start, Next),
    \+ member(Next, Path),
    Next_Max is Max -1,
    depthfirst_DL_aux(Next, [Next|Path], Solution, Next_Max).
```

% Iterative deepening search

```
depthfirst_IT(Start, Solution) :-
    depthfirst_IT_aux(Start,Solution, 0).
```

```
depthfirst_IT_aux(Start,Solution,Max) :-  
    depthfirst_DL_acyclic(Start,Solution, Max),!.
```

```
depthfirst_IT_aux(Start,Solution,Max) :-  
    not(depthfirst_DL_acyclic(Start,Solution, Max)),  
    New_Max is Max + 1,  
    depthfirst_IT_aux(Start,Solution,New_Max).
```

% breadth-first search

```
breadthfirst(Start, Solution) :-  
    breadthfirst_aux([[Start]], Temp_Solution),  
    reverse(Temp_Solution, Solution).
```

```
breadthfirst_aux([[Node|Rest]|_],[Node|Rest]) :-  
    goal(Node),!.
```

```
breadthfirst_aux([Path|Rest],Solution):-  
    extend(Path, NewPath),  
    append(Rest, NewPath, FinalPath),  
    breadthfirst_aux(FinalPath, Solution).
```

```
extend([Node|Path], NewPath) :-  
    findall([NewNode,Node|Path],(move(Node,NewNode), \+  
member(NewNode,[Node|Path])),NewPath).
```